

实验六 并行I/O接口的扩展

一、 实验目的

1. 了解并行 I/O 接口的扩展方法;
2. 掌握可编程接口芯片 8255A 的工作原理、编程方式和使用方法。

二、 实验器材

微机一台，TPC—H通用微机接口实验箱一个。

三、 实验原理

1. 并行I/O口的扩展方法

微机系统在实际应用中常常需要扩展I/O端口。

扩展并行I/O口主要有三种方法:

(1) 通过通用的 I/O 扩展芯片实现。如 Intel 公司的 8255A、8155; Zilog 公司的 PIO 等。

(2) 通过TTL、CMOS锁存器、缓冲器芯片实现。如 74LS377、74LS273、74LS244、74LS245 等。

(3) 通过串行通信口扩展并行I/O口。

本实验中采用 8255A 扩展 I/O 端口。

2. 采用 8255A 扩展 I/O 端口

8255A 是一种被广泛应用的可编程并行 I/O 接口芯片。

2.1 8255A 的引脚和内部结构




图 6.1 8255A 内部结构框图




图 6.2 8255A引脚图

表 6.1 地址选择信号

A0	A1	寄存器
0	0	端口A数据寄存器
0	1	端口B数据寄存器
1	0	端口C数据寄存器
1	1	控制寄存器


表 6.2 8255A 端口操作表


\overline{CS}	\overline{RD}	\overline{WR}	A1	A0	操作
0	0	1	0	0	读A口
0	0	1	0	1	读B口
0	0	1	1	0	读C口
0	1	0	0	0	写A口
0	1	0	0	1	写B口
0	1	0	1	0	写C口
0	1	0	1	1	写控制口
0	1	1	×	×	数据总线为三态
0	0	1	1	1	非法操作
1	×	×	×	×	数据总线为三态

2.2 8255A 的编程控制字

8255A具有三种基本工作方式，即方式0、方式1、方式2。各端口的工作方式及输入输出方向都由方式控制字设定（即编程）。8255A的编程是通过向其控制口送控制字来实现的，8255A的控制字有方式控制字和C口置位 / 复位控制字两种。

(1) 方式控制字





(b) C口置位/复位控制字

图 6.3 8255A的控制字格式

方式控制字的格式如图 6.3 (a) 所示。D7 位为标志位，必须为 1。C口可分为上半部 (PC7~PC4) 和下半部 (PC3~PC0)，上半部属于A组，下半部属于B组。

例如 8255A的A口为方式 0 输入，B口为方式 0 输出，C口上半部为方式 0 输出，下半部为方式 0 输入，设 8255A的控制口地址为 28BH，则用以下指令来编程：

```
MOV DX, 28BH
MOV AL, 10010001B      ;控制字送入累加器
OUT DX, AL             ;控制字送入 8255A
```

(2) C口置位 / 复位控制字

C口置位 / 复位控制字格式如图 6.3 (b) 所示，通过此控制字可以对C口的任意一位置“1”或清“0”。C口置位 / 复位控制字也是送入 8255A的控制寄存器，而不是 C 口，一次只能对C口的一位操作。

例如，将C口的PC4 置 1 用以下指令：

```
MOV DX, 28BH
MOV AL, 00001001B      ;C口置位 / 复位控制字送入A
OUT DX, AL             ;控制字送入 8255A
```

2.3 8255A 的工作方式

(1) 方式 0

方式 0 是一种基本I/O方式。在这种工作方式下，三个端口都可由程序设定为输入或输出，这种方式不使用联络信号，其基本功能为：

- ①两个 8 位端口 (A、B) 和两个 4 位端口 (C)。
- ②任一个端口可以作为输入或输出。
- ③输出锁存，输入不锁存。
- ④在方式 0 时，各端口都可以作为数据端口，CPU用简单的输入或输出指令来进行读或写。

同时，方式 0 也可以作为查询方式输入/输出的接口，此时端口A和B可以作为数据口，而端口C的某些位可以用来传送控制或状态信号。

(2) 方式 1

方式 1 是一种选通I/O方式。在这种方式下，端口A和B可作为数据端口，但C口的某

些位要用来传送控制或状态信号，C口的其它位仍可工作于方式 0。方式 1 的基本功能为：

- ①用作一个或两个选通端口。
- ②每个选通端口包含有：8 位数据端口，3 条控制线，提供中断逻辑。
- ③任一端口可输入或输出。
- ④若只有一个端口工作于方式 1，余下的 13 位可以工作于方式 0。
- ⑤若两个端口工作于方式 1，C口余下 2 位可以工作于方式 0。

下面分方式 1 输入和方式 1 输出分别介绍。

(i) 方式 1 输入

当A口或B口工作于方式 1 输入时，端口分配如图 6.4 所示。PA或PB作为输入数据口，C口的某些位作为它们的联络信号，各信号的意义为：

- ① \overline{STB} (Strobe) 选通输入信号，低电平有效，这是由输入设备提供的选通信号，当其有效时，把输入设备送来的数据送入输入锁存器。
- ②IBF (Input Buffer Full) 输入缓冲器满信号，高电平有效，它是 8255A送至外设的信号。当其有效时表示数据已输入至输入寄存器，它是由 \overline{STB} 信号下降沿将其置位（高电平）的。当CPU从 8255A中取走数据时， \overline{RD} 信号上升沿使IBF复位。




图 6.4 方式 1 输入组态

③INTR (Interrupt Request) 中断请求信号，高电平信号，它是 8255A的一个输出信号，用于向 CPU产生中断请求，要求CPU服务，当 \overline{STB} 、IBF和 INTE (中断允许)都为高电平时，INTR置位，而由 \overline{RD} 信号的下降沿清除。

④INTE (Interrupt Enable) 中断允许信号，用户可以通过对PC4 的置位/复位来控制INTE_A。通过对PC2 的置位/复位来控制INTE_B，INTE=1 时表示允许中断。

(ii) 方式 1 输出

在A口或B口用于方式 1 输出时。端口分配如图 6.5 所示。信号的时序关系可以参考图 6.6。各信号的意义为：

① \overline{OBF} (Output Buffer Full) 输出缓冲器满信号，低电平有效，8255A输出至外设的信号。其有效时，表示CPU已把数据输出到 8255A的端口，外设可以从 8255A的端口获得此数据输出。它由CPU的 \overline{WR} 上升沿置为有效，由外设的 \overline{ACK} 上升沿使其恢复为高电平。

② \overline{ACK} (Acknowledge) 外设响应信号，低电平有效。 \overline{ACK} 低电平时，表示外设已从 8255A 的端口获得数据输出，向 8255A 要下一个数据。




图 6.5 方式 1 输出组态

③ INTR 中断请求信号，高电平有效。当 \overline{ACK} 、 \overline{OBF} 和 INTE 都为高电平时，INTR 置位，而 CPU 的 \overline{WR} 信号使其复位（低电平）。

④ INTE 中断允许信号，PC6 的置位/复位控制 INTE_A ，PC2 的置位/复位控制 INTE_B 。




图 6.6 8255 方式 1 输入时序图 (单位 ns)




图 6.6 (续) 8255 方式 1 输出时序图 (单位 ns)

(3) 方式 2

方式 2 是一种双向 I/O 方式, 只有端口 A 具有这种工作方式, 其基本功能为:

- ① 一个 8 位双向数据端口 (A) 和一个 5 位控制端口 (C)。
- ② 输入和输出锁存。
- ③ 5 位控制端口用作端口 A 的状态和控制信息。

方式 2 的端口分配如图 6.7 所示, 各信号的意义为:

- ① INTR 中断请求, 输入和输出都可向 CPU 发出中断请求。
- ② INTE₁ 输出中断允许位, PC6 的置位/复位控制。




图 6.7 方式 2 的组态

- ③ INTE₂ 输入中断允许位, PC4 的置位/复位控制。

其他信号意义与方式 1 相同

端口 A 工作于方式 2 时, 端口 B 可以工作于方式 0 或者方式 1, 若工作于方式 1, 则 C 口的 8 根线全部作为联络信号使用。

四、 实验内容

(说明: 本实验中 8255A 的片选接到 288H~28FH 端子上, 各口地址分别为, A 口地址 IO 基地址+8, B 口地址 IO 基地址+9, C 口地址 IO 基地址+0AH, 控制口地址 IO 基地址+0BH, 示例代码中用 288H~28BH 代替)

1. 8255A 工作于方式 0, A 口输出, 接七段数码管; C 口输入, 接开关 K0-K3, 用七段数码管循环显示从 C 口低 4 位输入的数据。程序运行中如有键按下就返回 DOS。

参考程序如下:

```
DATA    SEGMENT
LED     DB 3FH, 06H, 5BH, 4FH, 66H, 6DH, 7DH, 07H
        DB 7FH, 67H, 77H, 7CH, 39H, 5EH, 79H, 71H

DATA    ENDS
CODE    SEGMENT
ASSUME  CS: CODE,  DS: DATA
START:  MOV AX, DATA
        MOV DS, AX           ;初始化 DS

        MOV DX, 28BH        ;设置 8255A 工作方式
        MOV AL, 89H
        OUT DX, AL

INOUT:  MOV DX, 28AH
        IN AL, DX           ;C 口输入


        AND AL, 0FH
        MOV BX, OFFSET LED
        XLAT                ;换码 AL←DS: [BX+AL]

        MOV DX, 288H
        OUT DX, AL          ;A 口输出字形码

        MOV AH, 1
        INT 16H
        JZ  INOUT           ;无按键转移
        MOV AH, 4CH         ;有按键则返回 DOS
        INT 21H

CODE    ENDS
END     START
```

主程序流程图如下



2. 8255A 工作于方式 0, A 口输出, C 口输入。从 C 口的高 4 位和低 4 位分别输入一个 8421BCD 码表示的十进制数, 求和后仍以 8421BCD 码十进制数形式在 LED 上显示。键盘上有键按下就返回 DOS。
3. 8255A 工作于方式 0, A 口输入, 接开关 K0-K7; C 口输出接 LED L0---L7。用 LED 以二进制数显示出 K0-K7 中置高电平的个数, 程序运行中如有键按下就返回 DOS。
4. 8255A 工作于方式 0, A 口输出数码管位码, C 口输出段码。在 S0 数码管上连续显示字符 0~F 后再在 S1 数码管上显示字符 0~F, 每个字符停留 1 秒, 循环显示。有按键返回到 DOS。

提示: 调用如下子程序或 BIOS 的等待功能, 适当调整参数可延时 1 秒

1) 延时子程序:

```
DELAY PROC NEAR
        MOV SI, 0200H
```

```
XX:    MOV CX, 0
YY:    LOOP YY
        DEC SI
        JNE XX
        RET
DELAY ENDP
```

2) BIOS 的等待功能调用:

```
MOV AH, 86H
MOV CX, 5
MOV DX, 0
INT 15H
```

5. 利用数码管实现两位数的秒表。

五、 实验报告附加要求

1. 画出实验内容 3 和 4 的程序简要流程图，注释实验内容 3 和 4 的主要程序段。